Module 23 Reguliere storingsrekening

Onderwerp	Machtreeksen, reguliere storingsrekening
Voorkennis	Gewone differentiaalvergelijkingen, reeksen
Expressies	`powsolve, tpsform, powpoly, subtract, powcreate, powdiff`
Bibliotheken	`powseries`
Zie ook	Module 12, 21

De Maple-bibliotheek `powseries` bevat een aantal procedures die gebruikt kunnen worden voor het creëren van formele machtreeksen[52] en om ermee te manipuleren.
Het gebruik van deze procedures zal worden gedemonstreerd aan de hand van het oplossen van gewone differentiaalvergelijkingen: in §23.2 worden oplossingen bepaald als een machtreeks in de onafhankelijke variabele en in §23.3 als machtreeks in een parameter (storing) waarvan de differentiaalvergelijking of de beginvoorwaarde afhangt.

23.1 Machtreeksen

We beschouwen lineaire stelsels:

$$\dot{\mathbf{x}} = A(t)\mathbf{x} + \mathbf{b}(t) \tag{23.1}$$

waarvoor zowel de elementen van de matrix $A(t)$ als de componenten van de vector $\mathbf{b}(t)$ reëel-analytische functies zijn. Dat betekent dat ze ontwikkelbaar zijn in een *machtreeks* (zeg rond $t = 0$) met een convergentie-interval, $|t| < R$ voor zekere $R > 0$,

$$A(t) = \sum_{k=0}^{\infty} A_k t^k \text{ en } \mathbf{b}(t) = \sum_{k=0}^{\infty} \mathbf{b}_k t^k.$$

We zoeken dan een oplossing in de vorm van een reeksontwikkeling:

$$\mathbf{x}(t) = \sum_{k=0}^{\infty} \mathbf{x}_k t^k\,, \tag{23.2}$$

dat wil zeggen de vectoren $\mathbf{x}_k$ worden gezocht zodat formele substitutie van (23.2) in (23.1) voldoet. Hierdoor is een oplossing verkregen als de reeks convergeert naar een differentieerbare functie.

[52] We spreken van *formele* machtreeksen als we ons niet om de convergentie ervan bekommeren.

23.2 Lineaire scalar differentiaalvergelijkingen

Beschouw het beginwaardeprobleem

$$\dot{x} = a(t)\,x, \quad x(0) = x_0, \tag{23.3}$$

waarin $a(t) = \sum_{k=0}^{\infty} a_k t^k$ convergent is voor $|t| < R$ $(\leq \infty)$. Substitutie van (23.2) in (23.3) levert de recursie-relatie als we gebruiken dat alle coëfficiënten van de machten van t nul moeten zijn.

$$(k+1)\,x_{k+1} = \sum_{j=0}^{k} a_{k-j} x_j.$$

Een afzonderlijke analyse moet aantonen dat de reeks sommeerbaar is en termsgewijs gedifferentieerd kan worden.

Voorbeeldopgave

Bepaal de oplossing van het beginwaardeprobleem

$$\frac{dx}{dt} = x, \quad x(0) = 2,$$

in de vorm van een reeksontwikkeling in t.

Voorbeeldsessie

```
> restart; with(powseries);
```

$[compose, evalpow, inverse, multconst, multiply, negative, powadd, powcos, powcreate, powdiff, powexp, powint, powlog, powpoly, powsin, powsolve, powsqrt, quotient, reversion, subtract, tpsform]$

```
> a := powsolve( diff(x(t),t)=x(t), x(0)=2 );
```

$$a := \mathbf{proc}(powparm)\ \ldots\ \mathbf{end\ proc}$$

De oplossing hebben we nu in de vorm van een machtreeks $\mathrm{x}(t) = \sum_{k=0}^{\infty} \mathrm{a}(k)\,t^k$. We kunnen verschillende informatie als volgt verkrijgen:

```
> a(10);
```

$$\frac{1}{1814400}$$

Dit leert ons wat de waarde is van de tiende coëfficiënt. De recursieformule waaraan de algemene coëfficiënt voldoet is op te vragen is als volgt:

```
> a(_k);
```

$$\frac{\mathrm{a}(\_k - 1)}{\_k}$$

Dit betekent dat de a(k)=a(k-1)/k. Let op het gebruik van de 'underscore' in het bovenstaand commando. Als we vervolgens een deel van de reeks willen zien dan kan dit via het commando "tpsform"(truncated power series form). Het eerste argument is de naam van de machtreeks, het tweede de naam van de variable die we willen gebruiken, het derde argument de orde.

```
> tpsform(a,t,5);
```

$$2 + 2\,t + t^2 + \frac{1}{3}\,t^3 + \frac{1}{12}\,t^4 + \mathrm{O}(t^5)$$

Toelichting

powsolve

tpsform

Het commando `powsolve` levert dus *niet* de machtreeks voor $x(t)$ zélf op, maar een procedure om de coëfficiënten te berekenen. De (eerste termen van de) reeksontwikkeling kunnen we te zien krijgen met het commando `tpsform`. Het herkennen van de machtreeks zal men nog zelf moeten doen, evenals het uitvoeren van de convergentietest. ◇

In het volgende voorbeeld laten we zien hoe ook een iets ingewikkelder probleem geheel met Maple kan worden opgelost.

Voorbeeldopgave

Beschouw de lineaire tweede orde homogene differentiaalvergelijking:

$$(t^2-1)\,\frac{d^2y}{dt^2} + 2t\frac{dy}{dt} - 2y = 0.$$

Bepaal de algemene oplossing als een reeks rond $t = 0$.

Voorbeeldsessie

```
> restart:  with(powseries):
> DV := (t^2-1)*diff(y(t),t,t) + 2*t*diff(y(t),t) - 2*y(t) = 0;
```

$$DV := (t^2-1)\,\left(\frac{d^2}{dt^2}\,\mathrm{y}(t)\right) + 2\,t\,\left(\frac{d}{dt}\,\mathrm{y}(t)\right) - 2\,\mathrm{y}(t) = 0$$

```
> a := powsolve( DV ):
> sum(a(k)*t^k,k=0..10);
```

$$C0 + C1\,t - C0\,t^2 - \frac{1}{3}\,C0\,t^4 - \frac{1}{5}\,C0\,t^6 - \frac{1}{7}\,C0\,t^8 - \frac{1}{9}\,C0\,t^{10}$$

```
> a(_k);
```

$$\frac{\mathrm{a}(\_k-2)\,(-3+\_k)}{\_k-1}$$

Hiermee hebben we feitelijk de algemene oplossing in de vorm van een reeks gevonden.

In dit geval kunnen we de som van de reeks expliciet bepalen. We doen dat voornamelijk om te laten zien welk soort manipulaties met een formele machtreeks mogelijk zijn.
Aangezien a_k alleen afhangt van a_{k-2} en bovendien $a_3 = 0$ (ga na!) volgt dat $t \to c\,t$ een oplossing moet zijn. Dit verifiëren we:

```
> eval( DV, y(t)=c*t );
```

$$0 = 0$$

Dit is Maple met ons eens. Nu krijgen we de algemene oplossing met C1=0 door te eisen dat de afgeleide in $t = 0$ gelijk is aan 0:

```
> a := powsolve( DV, D(y)(0)=0 ):
> sum( a(k)*t^k, k=0..10 );
```

$$C0 - C0\,t^2 - \frac{1}{3}\,C0\,t^4 - \frac{1}{5}\,C0\,t^6 - \frac{1}{7}\,C0\,t^8 - \frac{1}{9}\,C0\,t^{10}$$

Het lijkt erop dat als we van deze reeks C0 aftrekken, delen door t, en vervolgens differentiëren naar t, er een meetkundige reeks moet komen te staan. Dit vereist enig werk. Eerst zetten we de constante om in een machtreeks, waarvan dus alléén de eerste coëfficiënt niet nul is, namelijk die constante. Vervolgens trekken we die machtreeks af van a. Het resultaat noemen we b.

```
> p := powpoly(C0, x):  b := subtract(a, p):
> sum( b(k)*t^k, k=0..10 );
```

$$-C0\,t^2 - \frac{1}{3}\,C0\,t^4 - \frac{1}{5}\,C0\,t^6 - \frac{1}{7}\,C0\,t^8 - \frac{1}{9}\,C0\,t^{10}$$

Nu willen we delen door t. Dit is het zelfde als een nieuwe machtreeks te definiëren waarvan de coëfficiënten een plaats verschoven zijn ten opzichte van de coëfficiënten van b. Dit doen we als volgt:

```
> powcreate( c(n) = b(n+1), c(0)=0 );
> tpsform(c,t,7);
```

$$-C0\,t - \frac{C0}{3}\,t^3 - \frac{C0}{5}\,t^5 + \mathrm{O}(t^7)$$

We differentiëren c, het resultaat noemen we dc:

```
> dc := powdiff(c):
> tpsform(dc,t,10);
```

$$-C0 - C0\,t^2 - C0\,t^4 - C0\,t^6 - C0\,t^8 + \mathrm{O}(t^{10})$$

Onze hypothese is nu dat $dc = -\frac{C0}{1-t^2}$ (geometrische reeks). We bewandelen nu de weg terug. De constante C0 nemen we voor het gemak maar even gelijk aan -1. We integreren $\frac{1}{10\,t^2}$, vermenigvuldigen het resultaat met t en trekken er 1 vanaf. Dat zou dan een oplossing moeten zijn.

```
> dc := 1/(1-t^2);
```

$$dc := \frac{1}{1-t^2}$$

```
> b := t*int(dc,t) - 1;
```

$$b := t\,\mathrm{arctanh}(t) - 1$$

```
> eval( DV, y(t)=b ):  simplify(%);
```

$$0 = 0$$

Conclusie: De algemene oplossing van de differentiaalvergelijking is:

```
> Y := c1*t + c2*b;
```

$$Y := c1\,t + c2\,(t\,\mathrm{arctanh}(t) - 1)$$

Toelichting

powpoly

In dit voorbeeld hebben we kennisgemaakt met nog een aantal procedures uit de bibliotheek `powseries`, namelijk `powpoly`, waarmee van `subtract` `powcreate` `powdiff`

een polynoom een formele machtreeks wordt gemaakt; `subtract`, om twee machtreeksen van elkaar af te trekken, `powcreate`, om een nieuwe machtreeks te maken door de n^{de} coëfficiënt te specificeren, en zo nodig een of meer begincoëfficiënten te geven, en ten slotte `powdiff`, waarmee een formele machtreeks term voor term wordt gedifferentieerd. ◇

23.3 Problemen met een kleine parameter

Als een vectorveld en/of de beginconditie glad (dat wil zeggen voldoende vaak differentieerbaar) afhangt van een kleine parameter dan is het mogelijk om de oplossing van het beginwaardeprobleem te ontwikkelen naar de kleine parameter.

Stelling Laat I een compact interval zijn dat 0 bevat en laat voor $t \in I$ $\quad x^\varepsilon$ oplossing zijn van

$$\begin{cases} \dot{x} &= f(x,t;\varepsilon) \\ x(0) &= \alpha(\varepsilon). \end{cases}$$

Veronderstel dat f en α differentieerbaar zijn naar ε.
Dan geldt voor $t \in I$

$$x^\varepsilon(t) = x_0(t) + \varepsilon x_1(t) + o(\varepsilon),$$

waarin x_0 de oplossing is van het *limietprobleem* (met $\varepsilon = 0$):

$$\begin{cases} \dot{x}_0 &= f(x_0,t;0) \\ x_0(0) &= \alpha(0) \end{cases}$$

en waarin x_1 voldoet aan:

$$\begin{cases} \dot{x}_1 &= f_x(x_0(t),t;0)\,x_1 + f_\varepsilon(x_0(t),t;0) \\ x_1(0) &= \alpha_\varepsilon(0) \end{cases}$$

(hierin is f_x de partiële afgeleide van f naar x enzovoort.)
Dit resultaat kan recursief worden toegepast, wat dan een Taylor-ontwikkeling van de oplossing in de kleine parameter oplevert.

Voorbeeldopgave

We beschouwen het beginwaardeprobleem:

$$\frac{dx}{dt} = x + \varepsilon x^2, \quad x(0) = 1.$$

We schrijven $x^\varepsilon(t) = x_0(t) + \varepsilon x_1(t) + \mathcal{O}(\varepsilon^2)$, en bepalen de functies x_0 en x_1.

Voorbeeldsessie

```
> restart;
> DV := diff(x(t),t)-x(t)-epsilon*x(t)^2;
```

$$DV := (\frac{d}{dt}\,\mathrm{x}(t)) - \mathrm{x}(t) - \varepsilon\,\mathrm{x}(t)^2$$

```
> xeps := add(x||j(t)*epsilon^j, j=0..3);
```

$$xeps := \mathrm{x0}(t) + \mathrm{x1}(t)\,\varepsilon + \mathrm{x2}(t)\,\varepsilon^2 + \mathrm{x3}(t)\,\varepsilon^3$$

```
> eval( DV, x(t)=xeps ):  p := collect(%,epsilon);
```

$$\begin{aligned} p := & -\mathrm{x3}(t)^2\,\varepsilon^7 - 2\,\mathrm{x2}(t)\,\mathrm{x3}(t)\,\varepsilon^6 + \left(-2\,\mathrm{x1}(t)\,\mathrm{x3}(t) - \mathrm{x2}(t)^2\right)\,\varepsilon^5 \\ & + (-2\,\mathrm{x0}(t)\,\mathrm{x3}(t) - 2\,\mathrm{x1}(t)\,\mathrm{x2}(t))\,\varepsilon^4 \\ & + \left((\tfrac{d}{dt}\,\mathrm{x3}(t)) - \mathrm{x3}(t) - 2\,\mathrm{x0}(t)\,\mathrm{x2}(t) - \mathrm{x1}(t)^2\right)\,\varepsilon^3 \\ & + \left((\tfrac{d}{dt}\,\mathrm{x2}(t)) - \mathrm{x2}(t) - 2\,\mathrm{x0}(t)\,\mathrm{x1}(t)\right)\,\varepsilon^2 \\ & + \left((\tfrac{d}{dt}\,\mathrm{x1}(t)) - \mathrm{x1}(t) - \mathrm{x0}(t)^2\right)\,\varepsilon + \left(\tfrac{d}{dt}\,\mathrm{x0}(t)\right) - \mathrm{x0}(t) \end{aligned}$$

We krijgen nu de vergelijking voor x_0 door te eisen dat de coëfficiënt van ε^0 gelijk is aan 0.

```
> eqn0:=coeff(p,epsilon,0);
```

$$eqn0 := \frac{d}{dt}\,\mathrm{x0}(t) - \mathrm{x0}(t)$$

```
> dsolve( {eqn0, x0(0)=1}, x0(t) ); assign(%):
```

$$\mathrm{x0}(t) = e^t$$

Met dit `assign`-statement bereiken we dat nu voortaan x0(t) gelijk is aan e^t. Dit hoeven we dan niet meer te substitueren. Vervolgens vinden we de vergelijking voor x_1 door te eisen dat de coefficient van ε gelijk is aan 0.

```
> eqn1:=coeff(p,epsilon,1);
```

$$eqn1 := (\frac{d}{dt}\,\mathrm{x1}(t)) - \mathrm{x1}(t) - (e^t)^2$$

```
> dsolve( {eqn1,x1(0)=0}, x1(t) );
```

$$\mathrm{x1}(t) = (e^t - 1)\,e^t$$

Een benadering voor de oplossing is dus: $\mathrm{x}(t) = e^t + \varepsilon\,\left(e^{2\,t} - e^t\right) + \mathrm{O}(\varepsilon^2)$.
We kunnen de differentiaalvergelijking ook exact oplossen en daarmee het antwoord controleren.

```
> oplossing := rhs( dsolve( {DV=0, x(0)=1}, x(t) ) );
```

$$oplossing := \frac{1}{-\varepsilon + e^{-t} + e^{-t}\,\varepsilon}$$

```
> expand(taylor(oplossing,epsilon,2));
```

$$e^t + \left((e^t)^2 - e^t\right)\,\varepsilon + \mathrm{O}(\varepsilon^2)$$

Dit is consistent met wat we hebben gevonden.

Opgave 23.1

Gebruik de machtreeks-substitutie om de volgende problemen op te lossen. Als een expliciete formule voor de coëfficiënten x_k gevonden kan worden, schrijf dan de gehele reeks op. Als dat niet kan, volsta met enkele termen. Vergelijk het antwoord (of de benadering hiervan) met de numeriek geïntegreerde oplossing grafisch.

(a) $(1-t)\dot{x} = t^2 - x$ (Wat is het convergentie-interval?).
(b) $\dot{x} = 2t^2 + 3x,\ x(0) = 1.$
(c) $\dot{x} = 1 - t + 2x,$ rond $t = 1.$ (Bepaal de oplossing ook exact.)
(d) $t\dot{x} = t^4 + x,$ rond $t = 0.$ (Wat is het convergentie-interval?).
(e) $\dot{x} = t - 2x,\ x(0) = 1/2.$
(f) $y'' + (t-1)y' + y = 0,$ rond $t = 2.$

Opgave 23.2

Beschouw de differentiaalvergelijking

$$(1-t^2)y'' - 2ty' + a(a+1)y = 0,$$

a reëel.

(a) Voor welke a heeft de differentiaalvergelijking een polynoom als oplossing?
(b) Bepaal de oplossing voor $a = 5$ die voldoet aan $y(0) = 0$ en $y'(0) = 3.$

Opgave 23.3

Beschouw de homogene differentiaalvergelijking

$$(t^2-1)y'' + 2ty' - 2y = 0.$$

(a) Bepaal de algemene oplossing als een reeks rond $t = 0.$
(b) Sommeer de gevonden reeks en bepaal het convergentie-interval.
(c) Verifieer door directe substitutie dat de in (b) gevonden functie een oplossing is, en geef aan voor welke t-waarden dat het geval is. Vergelijk deze t-waarden met het in (b) gevonden convergentie-interval.
(d) Bepaal de algemene oplossing van de inhomogene differentiaalvergelijking $(t^2-1)y'' + 2ty' - 2y = \frac{1}{2}t^2 - 4.$

Opgave 23.4

Gegeven is de differentiaalvergelijking

$$(1+t^2)\ddot{y} + t\dot{y} - y = 0.$$

(a) Stel $y(t) = \sum_{n=0}^{\infty} c_n t^n$. Laat zien dat $c_2 = \frac{c_0}{2}$ en $c_3 = 0$.
(b) Hoeveel lineair onafhankelijke oplossingen heeft deze vergelijking?
(c) Bepaal de reeksontwikkeling tot en met orde 4.

Opgave 23.5

Gegeven is het beginwaardeprobleem

$$\begin{cases} (1+2\varepsilon)\,\ddot{x} + x = \varepsilon x^2 \\ x(0) = \varepsilon, \quad \dot{x}(0) = 1 \end{cases}$$

Schrijf de oplossing als machtreeks in ε:

$$x^{\varepsilon}(t) = x_0(t) + \varepsilon\, x_1(t) + \mathcal{O}(\varepsilon^2).$$

(a) Bepaal het beginwaardeprobleem waaraan x_0 voldoet en los dit op.
(b) Bepaal het beginwaardeprobleem waaraan x_1 voldoet en los dit op.

Opgave 23.6

Schrijf voor de volgende problemen de oplossing in de vorm

$$x^{\varepsilon}(t) = x_0(t) + \varepsilon\, x_1(t) + \mathcal{O}(\varepsilon^2)$$

en bepaal x_0 en x_1.

(a) $\begin{cases} \ddot{x} + \varepsilon\left(\frac{1}{3}(\dot{x})^3 - \dot{x}\right) + x = 0 \\ x(0) = \alpha \in \mathbb{R}, \qquad \dot{x}(0) = 0. \end{cases}$

(b) $\begin{cases} \ddot{x} - \varepsilon x\dot{x} + x = 0 \\ x(2\pi) = x(0), \qquad \dot{x}(2\pi) = \dot{x}(0). \end{cases}$

Opgave 23.7

Beschouw het beginwaardeprobleem

$$\begin{aligned} \dot{r} &= \alpha r + c r^3 \\ \dot{\theta} &= 1 \\ r(0) &= \varepsilon \\ \theta(0) &= 0. \end{aligned}$$

Schrijf de oplossing als een machtreeks in ε en bepaal bovendien de relatie tussen α en c zo dat de oplossing 2π-periodiek is.